

Estymacja modeli ARIMA przy użyciu Staty  
oraz  
Integracja i kointegracja

Grzegorz Ogonek  
KSiE WNE UW  
26.02.2005

\* Materiały opracowano w wersji 7 Staty. Tam gdzie zauważyłem rozbieżności z kolejną wersją podaję odpowiedniki komend dla Staty 8.

Budowa modelu ARIMA dla szeregu czasowego PPI (Producer Price Index) dla Polski dla okresu 1997:1 – 2004:12

Zacznijmy od polecenia:

```
set matsize 200
```

Zbiór danych: cpi\_ppi.xls

W zbiorze zamieniono przecinki na kropki, żeby można było przekleić wartości do staty.

Zbiór zawiera także CPI (Consumer Price Index).

Zmienna czasu to obs. Wpisano ją z błędem – wstawiając literę q między miesiąc a rok. Nie stanowi to jednak żadnego problemu dla staty.

Utwórzmy zmienną t:

```
gen t=monthly(obs,"my",2040)
```

Trzeci argument musimy podać, gdyż rok w zmiennej obs zadany jest dwiema cyframi. 2040 oznacza, że liczby od 0 do 40 będą traktowane jako lata 2000-2040, a liczby 41-99 jako 1941-1999.

Zmieńmy format wyświetlania zmiennej t, żeby nadać jej intuicyjną interpretację:

```
format t %tm
```

Zadeklarujmy nasz zbiór jako zbiór szeregów czasowych:

```
tsset t
```

Obejrzyjmy badaną zmienną:

```
graph ppi t, s(i) c(l)
```

```
(scatter ppi t, s(i) c(l) albo twoway (tsline=ppi) t) dla Staty8
```

opcja c(l) karze połączyć punktu wykresu odcinkami; s(i) ukrywa kropki

Ściągnijmy od razu ado-file „arimafit” -> otworzyć Stata Viewer (np. ctrl+3), search (wcisnąć s), zaznaczyć „search net resources”, wpisać „arimafit”. Jak się nie uda można używać pliku ic.do do wyświetlania Kryteriów Informacyjnych (wyświetla to samo, ale podzielone przez liczbę obserwacji). Zainstalujmy też ado-file „kpss” oraz „tauprob”.

## Ustalanie p, d, q w modelu ARIMA

Parametry p, d, q w modelu ARIMA ustalić można na trzy sposoby: korzystając z procedury Boxa-Jenkinsa, korzystając z Kryteriów Informacyjnych, korzystając z metodologii od ogólnego do szczegółowego.

Procedura Boxa-Jenkinsa (z czasów mniejszej mocy obliczeniowej komputerów)

Bazuje na „ocenie wzrokowej” (visual inspection, eyeball tests :))

Staramy się odnaleźć w wykresach funkcji ACF i PACF charakterystyczne wzorce – patrz str. 39-43 pliku wykład16\_17.pdf i na tej podstawie ustalamy p i q

Jeśli któraś z funkcji nie maleje prawie w ogóle to najwyraźniej zmienna objaśniana jest niestacjonarna i należy ją zróżnicować (zwiększyć parametr q o 1).

Wykorzystanie Kryteriów Informacyjnych. Zaczynamy od oszacowania największego sensownego modelu. Szacujemy modele dla wszystkich par p, q takich, że p, q są nie większe od tych wziętych na początku, np. wystartowaliśmy z modelu ARMA(4,4). Zapisujemy AIC/SIC. Szacujemy ARMA(4,3) (3,4) (3,3) (4,2) (2,4) (3,2) (2,3) ... (1,0) (0,1) (0,0) za każdym razem wyświetlając AIC/SIC. Wygrywa model z najmniejszymi wartościami kryteriów.

Procedura od ogólnego do szczegółowego (general-to-specific)

Zaczynamy od największego sensownego modelu (model ogólny). Wyrzucamy jeden ze składników AR lub MA. Badamy testem LR (Likelihood Ratio) czy okroiliśmy model za bardzo (H0 odrzucona) czy też to przejście jest uzasadnione (H0 nie odrzucona). W tym drugim przypadku następny składnik AR lub MA i badamy testem LR (względem modelu ogólnego a nie poprzednio oszacowanego), itd.

Jak to zrobić w Stacie?

Wyświetlmy korelogram dla naszej zmiennej objaśnianej (ppi)


corrgram ppi

LAG	AC	PAC	Q	Prob>Q	-1	0	1	-1	0	1
					[Autocorrelation]			[Partial Autocor]		
1	0.9584	0.9604	113	0.0000	-----					-----
2	0.9010	-0.5062	213.73	0.0000	-----			----		
3	0.8413	0.1385	302.3	0.0000	-----				-	
4	0.7756	-0.2809	378.23	0.0000	-----			--		
5	0.7053	0.0578	441.56	0.0000	-----					
6	0.6294	-0.2620	492.44	0.0000	-----			--		
7	0.5493	0.0874	531.54	0.0000	-----					
8	0.4676	0.0362	560.12	0.0000	---					
9	0.3865	0.0615	579.82	0.0000	---					
10	0.3102	-0.1111	592.63	0.0000	--					
11	0.2351	0.0718	600.05	0.0000	-					
12	0.1709	0.2539	604.01	0.0000	-			--		
13	0.1310	0.1187	606.36	0.0000	-					
14	0.0965	-0.1634	607.64	0.0000				-		
15	0.0630	0.0089	608.19	0.0000						
16	0.0357	-0.0520	608.37	0.0000						
17	0.0128	0.0796	608.4	0.0000						
18	-0.0017	-0.0670	608.4	0.0000						
19	-0.0097	0.0023	608.41	0.0000						
20	-0.0168	-0.0081	608.45	0.0000						
21	-0.0217	0.1162	608.52	0.0000						
22	-0.0193	-0.0387	608.58	0.0000						


23	-0.0122	-0.0412	608.6	0.0000		
24	-0.0033	0.2294	608.6	0.0000		-

Kolumna 2 i 3 to wartości funkcji ACF i PACF. Kolumna Q zawiera statystykę Q (Ljung-Boxa) do łącznego testowania autokorelacji rzędu od 1 do k. Tą samą wartość dla każdego k można uzyskać komendą `wntestq ppi, lags(k)`. (white noise test Q).  
Funkcje ACF i PACF możemy też wyświetlić poleceniami:

`ac ppi`



`pac ppi`



Jest to o tyle wygodniejsze, że stała użyje trybu graficznego i dostajemy też na wykresie przedziały ufności dla poszczególnych wartości.

Wg procedury Boxa-Jenkinsa moglibyśmy zacząć procedurę od oszacowania modelu Ze składnikami AR(1) AR(2) AR(4) AR(6) oraz MA od 1 do 6.

```
arima ppi, ar(1 2 4 6) ma(1 2 3 4 5 6)
```

Równie dobrze moglibyśmy zacząć od modelu ARMA(6,6) czyli ARIMA(6,0,6)

```
arima ppi, ar(1 2 3 4 5 6) ma(1 2 3 4 5 6)
```

co można też zapisać:

```
arima ppi,ar(1/6) ma(1/6)
```

lub

```
arima ppi,arima(6,0,6)
```

Albo potraktować wykres ACF jako wygasanie wykładnicze i zrezygnować ze wszystkich składników MA.

```
arima ppi,ar(1/6)
```

Jest tu pewna arbitralność. O ile procedura ma znamiona systematyczności – jest dozwolona.

### **Metoda Boxa-Jenkinsa**

Procedujemy dalej obierając za model wyjściowy

```
arima ppi,ar(1 2 4 6) ma(1 2 3 4 5 6)
```

Przechwyćmy reszty z tego modelu:

```
predict u0,residuals
```

i sprawdzimy czy w ich korelogramie można dostrzec jakiś wzorec. Jeśli nie – to mamy dobry model wyjściowy, jeśli widać wzór – najwyraźniej model wyjściowy trzeba rozszerzyć. Pomińmy z przyczyn technicznych fakt, że składniki MA(12), AR(12), AR(24) oraz składniki AR wyższego rzędu najprawdopodobniej należałoby włączyć do modelu. (stata ma trudności z osiągnięciem zbieżności przy szukaniu maksimum funkcji wiarygodności).

```
corrgram u0
```

Reszty są czyste.

Usuńmy z modelu składnik MA(6). Szacujemy:

```
arima ppi,ar(1 2 4 6) ma(1 2 3 4 5)
```

Przechwyćmy reszty i zbadajmy ich czystość:

```
predict u1,re
```

```
corrgram u1
```

Reszty są czyste.

I tak dalej. Proponowana dalsza ścieżka redukcji to: usunięcie MA(5),

```
arima ppi,ar(1 2 4 6) ma(1 2 3 4)
```

```
predict u2,re
```

```
corrgram u2
```

MA(4),

```
arima ppi,ar(1 2 4 6) ma(1 2 3)
```

```
predict u3,re
```

```
corrgram u3
```

MA(3),

```
arima ppi,ar(1 2 4 6) ma(1 2)
predict u4,re
corrgram u4
```

AR(6),

```
arima ppi,ar(1 2 4) ma(1 2)
predict u5,re
corrgram u5
```

AR(4),

```
arima ppi,ar(1 2) ma(1 2)
predict u6,re
corrgram u6
```

MA(2),

```
arima ppi,ar(1 2) ma(1)
predict u7,re
corrgram u7
```

AR(2),

```
arima ppi,ar(1) ma(1)
predict u8,re
corrgram u8
```

MA(1),

```
arima ppi,ar(1)
predict u9,re
corrgram u9
```

Po każdej redukcji patrzymy na korelogram reszt z modelu.

Dla ostatniego modelu w resztach pojawia się wzorzec, co oznacza, że ostatni krok redukcji był nieuprawniony – usunęliśmy w nim istotne składniki AR bądź MA.

Wróćmy do ARMA(1,1) i usuńmy AR(1). Znowu, reszty nie są czyste. Zatem procedura Boxa-Jenkinsa doprowadziła nas do modelu ARIMA(1,0,1), gdyż dalsza redukcja nie jest już możliwa.

### **Procedura z wykorzystaniem Kryteriów Informacyjnych**

Oszacujmy ponownie model wyjściowy dodając na początku komendy „quietly” – zapobiegając wyświetleniu wyniku:

```
quietly arima ppi,ar(1 2 4 6) ma(1 2 3 4 5 6)
```

Wyświetlmy wartości Kryteriów AIC i SIC komendą arimafit albo uruchamiając do-file ic (czyli komenda: do ic) plik ic.do musi być w katalogu domyślnym Staty.

Oszacujemy wszystkie użyte dotychczas modele za każdym razem dodając „quietly” i uruchamiając arimafit lub „do ic”. Dorzucimy dowolne inne specyfikacje modelu, np. ARMA(1,2), (3,1), (3,3), (1,3). Uwaga! Kryteria mogą wskazywać na dwa różne modele.

### **Procedura od ogólnego do szczegółowego.**

Znów szacujemy:

```
quietly arima ppi,ar(1 2 4 6) ma(1 2 3 4 5 6)
```

traktując go jako nasz model ogólny.

Zapiszmy uzyskaną w nim minimalną wartość logarytmu funkcji wiarygodności:

```
lrtest, saving(0) force
```

(w Stacie8 nie trzeba dodawać opcji force)

Stata w podejrzany sposób szuka minimum globalnego zmieniając metodę poszukiwania i najwyraźniej czasami utyka na minimum lokalnym. Dlatego też trzeba ją zmuszać (force) do zapisywania uzyskanego z modelu minimum funkcji wiarygodności. Z uwagi na tą wadę czasami test LR wychodzi bez sensu (tzn. statystyka  $\chi^2 < 0$ )

Podążajmy poprzednio ustaloną ścieżką redukcji po każdej estymacji stosując komendę:

```
lrtest, force
```

(w Stacie8 bez opcji force)

Nieodrzućenie  $H_0$  świadczy o łącznej nieistotności wszystkich usuniętych dotychczas składników.

Odrzućenie  $H_0$  oznacza, że zbyt mocno okroiliśmy model. Ostatni krok należy wówczas cofnąć i pójść w innym kierunku z redukowaniem modelu.

### **Prognozowanie z modelu ARIMA**

Oszacujmy model ARIMA(1,0,1)

Polecenie:

```
predict yhat, xb
```

wygeneruje zmienną yhat zawierającą prognozy (można użyć dowolnej nazwy własnej; yhat, czyli „y z daszkiem”). Tam gdzie to możliwe Stata liczy prognozy na 1 okres do przodu (one-step ahead forecasts). Gdy dochodzi do końca próby zaczyna tworzyć prognozę dynamiczną – opóźnione wartości zmiennej objaśnianej zastąpione zostają ich prognozami, czyli prognozy na następny okres liczone są z wykorzystaniem prognoz dla poprzednich okresów.

Możemy też zadać Stacie, aby liczyła prognozę dynamiczną zaczynając od dowolnego okresu, np. 2004m3 (przed wejściem do UE):

```
predict yhat1, xb dynamic(m(2004,3))
```

Dołączmy do prognoz punktowych ich przedział ufności. W tym celu wyprognozujmy odchylenie standardowe prognozy (nazwijmy je „odch”) dla każdego okresu prognozy:

```
predict odch, mse
```

( predict odch1, mse dynamic(m(2004,3)) dla prognozy dynamicznej dla 2004m3)

i utwórzmy dwie nowe zmienne – górny i dolny koniec 95% przedziału ufności:

```
gen cu=yhat+2*odch
gen cl=yhat-2*odch
(gen cu1=yhat1+2*odch1
gen cl1=yhat1-2*odch1 dla prognozy dynamicznej)
```

Zilustrujmy to wykresem:

```
graph ppi xb cl cu t, s(i i i) c(1111)
(polecenie scatter albo twoway w Stacie8)
```

Prognoza dla pierwszego okresu musi być nieudana, więc obetnijmy próbę prezentowaną na wykresie (weźmy od 2. obserwacji do ostatniej):

```
graph ppi xb cl cu t in 2/-1, s(i i i) c(1111)
(polecenie scatter albo twoway w Stacie8)
```

### Stopień zintegrowania (d) – formalne testowanie

Test Dickey-Fullera (DF) uruchamia się poleceniem

```
dfuller ppi
Dickey-Fuller test for unit root                               Number of obs   =           95
```

Test Statistic	Interpolated Dickey-Fuller		
	1% Critical Value	5% Critical Value	10% Critical Value
Z(t)	-3.517	-2.894	-2.582

\* MacKinnon approximate p-value for Z(t) = 0.2030

Jeśli chcemy użyć rozszerzonego testu DF, czyli ADF musimy dodać opcję lags(k), gdzie k to ilość zastosowanych rozszerzeń.

Dodatkowo opcja reg wyświetli oszacowanie parametrów regresji służącej do wyliczania statystyki DF (statystyka t przy l.ppi)

### Procedura testowania integracji

Policzyć statystykę DF dla dużej sensownej liczby rozszerzeń. Sprawdzić testem Breuscha-Godfrefya obecność autokorelacji (najlepiej dla kilku rzędów). W przypadku występowania autokorelacji musimy dołożyć kolejne rozszerzenia. Jeśli brak autokorelacji to zmniejszamy liczbę rozszerzeń i znów badamy czy nie pojawiła się autokorelacja. Jeśli się pojawi dodajemy z powrotem ostatnie rozszerzenie i z takiego modelu odczytujemy najefektywniejsze oszacowanie parametru przy l.ppi i jego statystykę t (czyli interesującą nas statystykę DF).

Do badania zmiennej ppi trzeba użyć 5 rozszerzeń. Otrzymany wówczas wniosek to jej niestacjonarność na 5% poziomie istotności.

```
dfuller ppi, lags(5)
Augmented Dickey-Fuller test for unit root                   Number of obs   =           90
```

	Test Statistic	----- 1% Critical Value	Interpolated Dickey-Fuller 5% Critical Value	----- 10% Critical Value
Z(t)	-2.502	-3.524	-2.898	-2.584

\* MacKinnon approximate p-value for Z(t) = 0.1150

bgodfrey, lags(1)  
Breusch-Godfrey LM statistic: .3779868 Chi-sq( 1) P-value = .5387

bgodfrey, lags(2)  
Breusch-Godfrey LM statistic: .6934929 Chi-sq( 2) P-value = .707

bgodfrey, lags(4)  
Breusch-Godfrey LM statistic: 2.699539 Chi-sq( 4) P-value = .6093

bgodfrey, lags(6)  
Breusch-Godfrey LM statistic: 6.052945 Chi-sq( 6) P-value = .4173

bgodfrey, lags(12)  
Breusch-Godfrey LM statistic: 13.47095 Chi-sq(12) P-value = .3358

Oznacza to, że nasze wcześniejsze uznanie szeregu ppi za szereg I(0) jest podważone testem ADF.

Badając stacjonarność cpi dochodzimy do testu ADF z 1 rozszerzeniem. Otrzymana statystyka DF to -2.538 wobec 5%-owej wartości krytycznej -2.907. Zmienna cpi jest zatem zintegrowana stopnia 1.

Skoro ppi i cpi są obie I(1) to możemy zbadać czy są one skointegrowane. Oszacujmy:

```
reg ppi cpi, nocons
```

Będzie to relacja długookresowa między tymi zmiennymi. Należy zachować ostrożność przy orzekaniu istotności regresorów w relacji długookresowej. Zauważmy, że obie występujące w niej zmienne są niestacjonarne. Jesteśmy więc narażeni na problem regresji pozornej. W takim przypadku przy testowaniu istotności nie wolno nam wykorzystać wyrzucanych przez Statę wartości p (p-values). Relację długookresową można też oszacować na podstawie modelu ADL wyliczając mnożniki długookresowe.

Zapiszmy reszty z tej regresji:

```
predict u_koint, re
```

i zbadajmy ich stacjonarność testem DF / ADF. Tym razem musimy użyć dla statystyki t dla l.u\_koint specjalnych wartości krytycznych zależnych od liczby szacowanych parametrów równania długookresowego. Są one stabilizowane na końcu książki Charemza, Deadman, *Nowa Ekonometria*. Można je też wyświetlić używając bezpośrednio po komendzie dfuller polecenia

```
tauprob c 2 r(Zt)
macro list S_1
```

„c” oznacza, że w regresji pomocniczej testu DF / ADF użyto stałej (gdyby to była stała i trend wpisalibyśmy „ct”; stała i trend kwadratowy – „ctt”). Jako drugi argument polecenia tauprob podajemy liczbę zmiennych w wektorze kointegrującym. Zamiast r(Zt) można ręcznie wpisać wartość otrzymanej statystyki DF.

```
. dfuller u_koint, lags(1)
```



Augmented Dickey-Fuller test for unit root      Number of obs =      79

	----- Interpolated Dickey-Fuller -----			
Test	1% Critical	5% Critical	10% Critical	
Statistic	Value	Value	Value	
Z(t)	-1.645	-3.539	-2.907	-2.588

\* MacKinnon approximate p-value for Z(t) = 0.4596

. bgodfrey

Number of gaps in sample: 1

Breusch-Godfrey LM statistic: 3.420096 Chi-sq( 1) P-value = .0644

. bgodfrey, lags(2)

Number of gaps in sample: 1

Breusch-Godfrey LM statistic: 4.989994 Chi-sq( 2) P-value = .0825

. bgodfrey, lags(3)

Number of gaps in sample: 1

Breusch-Godfrey LM statistic: 6.697403 Chi-sq( 3) P-value = .0822

. bgodfrey, lags(4)

Number of gaps in sample: 1

Breusch-Godfrey LM statistic: 8.082115 Chi-sq( 4) P-value = .0886

. bgodfrey, lags(5)

Number of gaps in sample: 1

Breusch-Godfrey LM statistic: 8.257671 Chi-sq( 5) P-value = .1426

. dfuller u\_koint, lags(1) (powtarzamy, żeby w pamięci staty znalazła się statystyka DF)

. tauprob c 2 r(Zt)

. macro list S\_1

S\_1: .7011781612201308 (liczba ta to asymptotyczna wartość p dla l.u\_koint)

Reszty okazują się być niestacjonarne (1 rozszerzenie, badanie ze stałą)

Regresja ppi na cpi i stałej również daje reszty niestacjonarne (2 rozszerzenia, badanie bez stałej (opcja nocons))

Zatem stwierdzamy brak kointegracji między ppi i cpi.

Gdyby reszty okazały się stacjonarne moglibyśmy oszacować model ECM komendą:

```
reg d.ppi l.u_koint d.cpi
```

dodając opóźnione zmienne objaśniane (czyli ld.ppi l2d.ppi l3d.ppi, itd.) dopóki z reszt z tego modelu nie zniknie autokorelacja.

Można tym sposobem dojść do modelu:

```
reg d.ppi l.u_koint d.cpi ld.ppi l2d.ppi l3d.ppi, nocons
```

Parametr przy opóźnionych resztach z modelu relacji długookresowej okazuje się nieistotny, a więc relacja kointegrująca nie pracuje, co jest zgodne z wcześniejszym ustaleniem, że brak kointegracji między ppi i cpi.

Stacjonarność możemy też badać przy użyciu testu KPSS. Różni się on tym od testu DF/ADF, że jego hipoteza zerowa mówi o stacjonarności (dokładniej: o stacjonarności po usunięciu trendu liniowego).

```
kpss ppi
```

KPSS test for ppi

Maxlag = 11 chosen by Schwert criterion  
Autocovariances weighted by Bartlett kernel

Critical values for H0: ppi is trend stationary

10%: 0.119 5% : 0.146 2.5%: 0.176 1% : 0.216

Lag order	Test statistic
0	1.23649
1	.631603
2	.430954
3	.331397
4	.272383
5	.233719
6	.206764
7	.187213
8	.172665
9	.161636
10	.153164
11	.146638

Na 5% poziomie istotności przyjęcie dowolnej liczby opóźnień kończy się odrzuceniem H0 (wszystkie statystyki testowe większe od 5% wartości krytycznej 0.146), czyli stwierdzeniem niestacjonarności szeregu ppi.

. kpss d.ppi

KPSS test for D.ppi

Maxlag = 11 chosen by Schwert criterion  
Autocovariances weighted by Bartlett kernel

Critical values for H0: D.ppi is trend stationary

10%: 0.119 5% : 0.146 2.5%: 0.176 1% : 0.216

Lag order	Test statistic
0	.11626
1	.079763
2	.070453
3	.064442
4	.060483
5	.057042
6	.054238
7	.05293
8	.052647
9	.052382
10	.052088
11	.052446

Z kolei zróżnicowana zmienna ppi jest według testu KPSS stacjonarna niezależnie od przyjętej liczby opóźnień (tzn. nie ma podstaw do odrzucenia hipotezy o jej stacjonarności).